

The following code is taken from <https://mathematica.stackexchange.com/questions/11936/differentiate-the-product-of-some-terms/156311#156311>, allowing differentiation of products of variable length:

```
In[22]:= Unprotect[Product];
D[Product[f_, iter_], p_, q_] := D[D[Product[f, iter], p], q]
D[Product[f_, iter_], p_] := With[{res = iD[Product[f, iter], p]}, res /; res != $Failed]
Protect[Product];
```

```
iD[Product[f_, iter_], {p_, n_Integer}] :=
  If[ListQ[p], $Failed, Nest[D[#, p] &, iD[Product[f, iter], p], n - 1]]
```

```
iD[Product[f_, iter_], p_List] := $Failed
iD[Product[f_, iter_], p_] := Product[f, iter] * D[Sum[Log[f], iter], p]
```

```
In[37]:= u[y_] := Csch[y/2]^2/4
```

```
In[41]:= H := Sum[Product[Sqrt[1 + u[y[i] - y[j]] / u[η μ]], {j, 1, i - 1}] *
  Product[Sqrt[1 + u[y[i] - y[j]] / u[η μ]], {j, i + 1, NN}] Cosh[p[i] μ], {i, 1, NN}]
```

```
In[42]:= H
```

$$\text{Out[42]} = \sum_{i=1}^{NN} \text{Cosh}[\mu p[i]] \left(\prod_{j=1}^{i-1} \sqrt{1 + \text{Csch}\left[\frac{1}{2} (y[i] - y[j])\right]^2 \text{Sinh}\left[\frac{\eta \mu}{2}\right]^2} \right) \prod_{j=i+1}^{NN} \sqrt{1 + \text{Csch}\left[\frac{1}{2} (y[i] - y[j])\right]^2 \text{Sinh}\left[\frac{\eta \mu}{2}\right]^2}$$

```
In[43]:= Simplify[D[H, {μ, 2}]] /. μ -> 0
```

$$\text{Out[43]} = \sum_{i=1}^{NN} \left(p[i]^2 + \sum_{j=1}^{i-1} \frac{1}{4} \eta^2 \text{Csch}\left[\frac{1}{2} (y[i] - y[j])\right]^2 + \sum_{j=i+1}^{NN} \frac{1}{4} \eta^2 \text{Csch}\left[\frac{1}{2} (y[i] - y[j])\right]^2 \right)$$