The following code is taken from https://mathematica.stackexchange.com/questions/11936/differentiate-the-product-of-some-terms/156311#156311, allowing differentiation of products of variable length:

In[1]:=
```
Unprotect[Product];
D[Product[f_, iter_], p_, q__] ^:= D[D[Product[f, iter], p], q]
D[Product[f_, iter_], p_] ^:= With[{res = iD[Product[f, iter], p]}, res /; res =!= $Failed]
Protect[Product];

iD[Product[f_, iter_], {p_, n_Integer}] :=
 If[ListQ[p], $Failed, Nest[D[#, p] &, iD[Product[f, iter], p], n - 1]]

iD[Product[f_, iter_], p_List] := $Failed
iD[Product[f_, iter_], p_] := Product[f, iter] * D[Sum[Log[f], iter], p]
```

In[8]:=
```
u[y_] := Csch[y / 2]^2 / 4
```

In[9]:=
```
H := Sum[Product[Sqrt[1 + u[y[i] - y[j]] / u[η μ]], {j, 1, i - 1}] *
    Product[Sqrt[1 + u[y[i] - y[j]] / u[η μ]], {j, i + 1, NN}] Cosh[p[i] μ], {i, 1, NN}]
```

In[10]:=
```
H
```

Out[10]=
$$\sum_{i=1}^{NN} \text{Cosh}[\mu\, p[i]] \left(\prod_{j=1}^{-1+i} \sqrt{1 + \text{Csch}\left[\frac{1}{2}(y[i]-y[j])\right]^2 \text{Sinh}\left[\frac{\eta\,\mu}{2}\right]^2} \right) \prod_{j=1+i}^{NN} \sqrt{1 + \text{Csch}\left[\frac{1}{2}(y[i]-y[j])\right]^2 \text{Sinh}\left[\frac{\eta\,\mu}{2}\right]^2}$$

In[11]:=
```
Simplify[D[H, {μ, 0}]] /. μ → 0
```

Out[11]=
```
NN
```

In[13]:=
```
Simplify[D[H, {μ, 1}]] /. μ → 0
```

Out[13]=
```
0
```

In[14]:=
```
Simplify[D[H, {μ, 2}]] /. μ → 0
```

Out[14]=
$$\sum_{i=1}^{NN} \left(p[i]^2 + \sum_{j=1}^{-1+i} \frac{1}{4}\eta^2 \text{Csch}\left[\frac{1}{2}(y[i]-y[j])\right]^2 + \sum_{j=1+i}^{NN} \frac{1}{4}\eta^2 \text{Csch}\left[\frac{1}{2}(y[i]-y[j])\right]^2 \right)$$